7. přednáška (rekurentní rovnice, tvorba vlastních balíčků a nápovědy)

Rekurentní rovnice

Základním příkazem pro řešení rekurentních rovnic v Maplu je příkaz rsolve.

Příklad:

```
> restart;
> rovnice:=a(n+1)=2*a(n);
```

$$rovnice := a(n+1) = 2\,a(n)$$

```
> rsolve(rovnice,a(n));
```

$$a(0)\,2^n$$

Vidíme, že řešení není určeno jednoznačně. Rovnici vyhovuje každá geometrická posloupnost s kvocientem 2.

Stejně jako u diferenciálních rovnic lze uvažovat i počáteční podmínky.

```
> reseni:=rsolve({rovnice,a(1)=2},a(n));
```

$$reseni := 2^n$$

```
> seq(reseni,n=1..20);
```

2, 4, 8, 16, 32, 64, 128, 256, 512, 1024, 2048, 4096, 8192, 16384, 32768, 65536, 131072, 262144, 524288, 1048576

```
>
```

Příklad:

```
> rsolve({a(n+2)=-a(n),a(1)=1,a(2)=0},a(n));
```

$$-\frac{1}{2}\,I\,I^n + \frac{1}{2}\,I\,(-I)^n$$

Vidíme, že jsme dostali řešení v komplexním tvaru.

```
> reseni:=%;
```

$$reseni := -\frac{1}{2}\,I\,I^n + \frac{1}{2}\,I\,(-I)^n$$

```
> reseni:=evalc(reseni);
```

$$reseni := \sin\left(\frac{n\,\pi}{2}\right)$$

```
> seq(reseni,n=1..20);
```

1, 0, -1, 0, 1, 0, -1, 0, 1, 0, -1, 0, 1, 0, -1, 0, 1, 0, -1, 0

```
>
```

Příklad (Fibonacciho posloupnost):

```
> rekurze:=a(n+2)=a(n+1)+a(n);
```

$$rekurze := \mathrm{a}(n+2) = \mathrm{a}(n+1) + \mathrm{a}(n)$$

```
> poc_podminky:=a(1)=1,a(2)=1;
```

$$poc\_podminky := \mathrm{a}(1) = 1,\ \mathrm{a}(2) = 1$$

```
> rsolve({rekurze,poc_podminky},a(n));
```

$$-\frac{\sqrt{5}\left(-\frac{\sqrt{5}}{2}+\frac{1}{2}\right)^{n}}{5}+\frac{\sqrt{5}\left(\frac{1}{2}+\frac{\sqrt{5}}{2}\right)^{n}}{5}$$

```
> fib:=unapply(%,n);
```

$$fib := n \to -\frac{1}{5}\sqrt{5}\left(-\frac{\sqrt{5}}{2}+\frac{1}{2}\right)^{n}+\frac{1}{5}\sqrt{5}\left(\frac{1}{2}+\frac{\sqrt{5}}{2}\right)^{n}$$

```
> seq(simplify(fib(n)),n=1..20);
```

1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233, 377, 610, 987, 1597, 2584, 4181, 6765

```
> fib(2012);
```

$$-\frac{\sqrt{5}\left(-\frac{\sqrt{5}}{2}+\frac{1}{2}\right)^{2012}}{5}+\frac{\sqrt{5}\left(\frac{1}{2}+\frac{\sqrt{5}}{2}\right)^{2012}}{5}$$

```
> simplify(%);
```

13603390990508312913794356816889555030655856590109461551416026171838973\
3\
744370141109639994229713198207950263844577433115794627700508717188822855\
835553736622437174620046963788023631658859908688734803690188424159333661\
852453869066061806368212533954401328122353600471216111212316006409108550\
298641872693817086759202279661351711900463742125846751792341308404992065\
744569826285199584565828884017594318466481540767085336318226 9

Příklad

Příklad (banka - viz 4. přednášku):

Představme si banku, do které vložíme e Kč (Eulerovo číslo). Po každém roce se z účtu odečte 1 Kč (vedení účtu) a zůstatek se vynásobí počtem let od založení účtu. Zjistěte, jak výhodná je nabídka, kterou nám banka dává. Kolik Kč budeme mít na účtu za 22 let ?

Řešení:

Je jasné, že se jedná o rekurzi $a_n = n\,(a_{n-1} - 1)$, kde $a_0 = \mathbf{e}$. a_n pak udává stav účtu po n letech od založení účtu.

```
> rekurze:=a(n)=n*(a(n-1)-1);
```

$$rekurze := \mathrm{a}(n) = n\,(\mathrm{a}(n-1) - 1)$$

```
> rsolve({rekurze,a(0)=exp(1)},a(n));
```

$$\Gamma(n+1)\left(\left(\sum_{n1=0}^{n-1} \frac{-n1-1}{\Gamma(n1+2)}\right) + \mathbf{e}\right)$$

```
> reseni:=unapply(%,n);
```

$$reseni := n \rightarrow \Gamma(1+n)\left(\left(\sum_{n1=0}^{n-1} \frac{-n1-1}{\Gamma(n1+2)}\right) + \mathbf{e}\right)$$

Explicitní vyjádření jsme nedostali v pěkném tvaru.

```
> reseni(22);
```

$$1124000727777607680000\left(\sum_{n1=0}^{21} \frac{-n1-1}{\Gamma(n1+2)}\right) + 1124000727777607680000\,\mathbf{e}$$

```
> evalf(%,50);
```

$$1.0453652129514255963919495415$$

```
>
```

Příkazem rsolve lze řešit i soustavy rekurentních rovnic.

Příklad:

Najděte řešení soustavy rekurentních rovnic $a_{n+1} = a_n + b_n$, $b_{n+1} = -a_n + b_n$ s počátečními podmínkami $a_1 = 1$, $b_1 = 1$.

```
> restart;
> rsolve({a(n+1)=a(n)+b(n),b(n+1)=-a(n)+b(n),a(1)=1,b(1)=1},{a(
  n),b(n)});
```

$$\left\{\mathrm{a}(n) = -\frac{1}{2} I\,(1+I)^n + \frac{1}{2} I\,(1-I)^n,\ \mathrm{b}(n) = \frac{(1+I)^n}{2} + \frac{(1-I)^n}{2}\right\}$$

```
> evalc(%);
```

$$\left\{\mathrm{a}(n) = \mathbf{e}^{(1/2\,n\ln(2))}\sin\left(\frac{n\,\pi}{4}\right),\ \mathrm{b}(n) = \mathbf{e}^{(1/2\,n\ln(2))}\cos\left(\frac{n\,\pi}{4}\right)\right\}$$

```
> reseni:=simplify(%);
```

$$reseni := \left\{\mathrm{a}(n) = 2^{\left(\frac{n}{2}\right)}\sin\left(\frac{n\,\pi}{4}\right),\ \mathrm{b}(n) = 2^{\left(\frac{n}{2}\right)}\cos\left(\frac{n\,\pi}{4}\right)\right\}$$

```
> a:=rhs(reseni[1]);
```

$$a := 2^{\left(\frac{n}{2}\right)} \sin\left(\frac{n\,\pi}{4}\right)$$

```
> b:=rhs(reseni[2]);
```

$$b := 2^{\left(\frac{n}{2}\right)} \cos\left(\frac{n\,\pi}{4}\right)$$

```
> seq([a,b],n=1..10);
```

$$[1, 1], [2, 0], [2, -2], [0, -4], [-4, -4], [-8, 0], [-8, 8], [0, 16], [16, 16], [32, 0]$$

```
>
```

Příklad:

Je dána matice $A = \begin{bmatrix} -2 & 2 \\ -5 & 4 \end{bmatrix}$. Najděte předpis pro prvky matice A^n, kde n je přirozené číslo.

Řešení:

Označme a_n, b_n, c_n, d_n prvky matice A^n (po řádcích). Pak jistě stačí vyřešit následující soustavu rekurentních rovnic:

```
> restart;
> rsolve({a(n+1)=-2*a(n)-5*b(n),b(n+1)=2*a(n)+4*b(n),c(n+1)=-2*c(n)-5*d(n),d(n+1)=2*c(n)+4*d(n),a(1)=-2,b(1)=2,c(1)=-5,d(1)=4},{a(n),b(n),c(n),d(n)});
```

$$\left\{ \mathrm{a}(n) = \frac{(1-I)^n}{2} - \frac{3}{2} I (1-I)^n + \frac{(1+I)^n}{2} + \frac{3}{2} I (1+I)^n,\ \mathrm{b}(n) = (1-I)^n I - (1+I)^n I,\right.$$
$$\left. \mathrm{c}(n) = -\frac{5}{2} I (1-I)^n + \frac{5}{2} I (1+I)^n,\ \mathrm{d}(n) = \frac{(1-I)^n}{2} + \frac{3}{2} I (1-I)^n + \frac{(1+I)^n}{2} - \frac{3}{2} I (1+I)^n \right\}$$

```
> evalc(%);
```

$$\left\{ \mathrm{a}(n) = \mathbf{e}^{(1/2\,n\ln(2))} \cos\left(\frac{n\,\pi}{4}\right) - 3\,\mathbf{e}^{(1/2\,n\ln(2))} \sin\left(\frac{n\,\pi}{4}\right),\ \mathrm{b}(n) = 2\,\mathbf{e}^{(1/2\,n\ln(2))} \sin\left(\frac{n\,\pi}{4}\right),\right.$$
$$\left. \mathrm{c}(n) = -5\,\mathbf{e}^{(1/2\,n\ln(2))} \sin\left(\frac{n\,\pi}{4}\right),\ \mathrm{d}(n) = \mathbf{e}^{(1/2\,n\ln(2))} \cos\left(\frac{n\,\pi}{4}\right) + 3\,\mathbf{e}^{(1/2\,n\ln(2))} \sin\left(\frac{n\,\pi}{4}\right) \right\}$$

```
> reseni:=simplify(%);
```

$$reseni := \left\{ \mathrm{a}(n) = 2^{\left(\frac{n}{2}\right)} \left(\cos\left(\frac{n\,\pi}{4}\right) - 3\sin\left(\frac{n\,\pi}{4}\right)\right),\ \mathrm{b}(n) = 2^{\left(1+\frac{n}{2}\right)} \sin\left(\frac{n\,\pi}{4}\right),\right.$$
$$\left. \mathrm{c}(n) = -5\,2^{\left(\frac{n}{2}\right)} \sin\left(\frac{n\,\pi}{4}\right),\ \mathrm{d}(n) = 2^{\left(\frac{n}{2}\right)} \left(\cos\left(\frac{n\,\pi}{4}\right) + 3\sin\left(\frac{n\,\pi}{4}\right)\right) \right\}$$

```
> a:=unapply(rhs(reseni[1]),n);
```

$$a := n \to 2^{(1/2\,n)} \left(\cos\left(\frac{1}{4} n \pi\right) - 3 \sin\left(\frac{1}{4} n \pi\right) \right)$$

```
> b:=unapply(rhs(reseni[2]),n);
```

$$b := n \to 2^{(1 + 1/2\,n)} \sin\left(\frac{1}{4} n \pi\right)$$

```
> c:=unapply(rhs(reseni[3]),n);
```

$$c := n \to -5\, 2^{(1/2\,n)} \sin\left(\frac{1}{4} n \pi\right)$$

```
> d:=unapply(rhs(reseni[4]),n);
```

$$d := n \to 2^{(1/2\,n)} \left(\cos\left(\frac{1}{4} n \pi\right) + 3 \sin\left(\frac{1}{4} n \pi\right) \right)$$

```
> M:=n->matrix([[a(n),b(n)],[c(n),d(n)]]);
```

$$M := n \to \begin{bmatrix} \mathrm{a}(n) & \mathrm{b}(n) \\ \mathrm{c}(n) & \mathrm{d}(n) \end{bmatrix}$$

Předpis pro n-tou mocninu matice A:

```
> M(n);
```

$$\begin{bmatrix} 2^{\left(\frac{n}{2}\right)} \left(\cos\left(\frac{n \pi}{4}\right) - 3 \sin\left(\frac{n \pi}{4}\right) \right) & 2^{\left(1 + \frac{n}{2}\right)} \sin\left(\frac{n \pi}{4}\right) \\ -5\, 2^{\left(\frac{n}{2}\right)} \sin\left(\frac{n \pi}{4}\right) & 2^{\left(\frac{n}{2}\right)} \left(\cos\left(\frac{n \pi}{4}\right) + 3 \sin\left(\frac{n \pi}{4}\right) \right) \end{bmatrix}$$

A^{10}:

```
> M(10);
```

$$\begin{bmatrix} -96 & 64 \\ -160 & 96 \end{bmatrix}$$

Kontrola:

```
> A:=matrix([[-2,2],[-5,4]]);
```

$$A := \begin{bmatrix} -2 & 2 \\ -5 & 4 \end{bmatrix}$$

```
> evalm(A^10);
```

$$\begin{bmatrix} -96 & 64 \\ -160 & 96 \end{bmatrix}$$

n-tá mocnina matice A:

```
> evalm(A^n);
```

$$\begin{bmatrix} -2 & 2 \\ -5 & 4 \end{bmatrix}^n$$

Vidíme, že n-tou mocninu Maple obecně nespočítal.
My jsme však obecný vztah (pomocí rekurentních rovnic) našli.

```
>
```

Užití rekurentních rovnic v kombinatorice.

Příklad:

Uvažujme "žebřík" skládající se z n čtverců (viz obrázek níže). Některé ze stran čtverců obarvíme. Uděláme to ale tak, aby v každém čtverci byla obarvená alespoň jedna strana. Určete (obecně) počet všech takových obarvení. Kolik je obarvení pro $n = 470$?



Řešení:

Označme a_n počet takových obarvení. Není příliš obtížné přijít na to, že pak platí rekurentní vztah $a_{n+2} = 7\,a_{n+1} + 4\,a_n$. Dále lze spočítat, že $a_1 = 15$ a $a_2 = 113$.

```
> restart;
> rsolve({a(n+2)=7*a(n+1)+4*a(n),a(1)=15,a(2)=113},a(n));
```

$$\left(\frac{8\sqrt{65}}{65}+1\right)\left(\frac{7}{2}+\frac{\sqrt{65}}{2}\right)^n+\left(-\frac{8\sqrt{65}}{65}+1\right)\left(\frac{7}{2}-\frac{\sqrt{65}}{2}\right)^n$$

```
> reseni:=unapply(%,n);
```

$$reseni := n \rightarrow \left(\frac{8\sqrt{65}}{65}+1\right)\left(\frac{7}{2}+\frac{\sqrt{65}}{2}\right)^n+\left(-\frac{8\sqrt{65}}{65}+1\right)\left(\frac{7}{2}-\frac{\sqrt{65}}{2}\right)^n$$

Dostali jsme obecný vztah pro hledaný počet obarvení.

```
> seq(simplify(reseni(n)),n=1..20);
```

15, 113, 851, 6409, 48267, 363505, 2737603, 20617241, 155271099, 1169366657, 8806650995, 66324023593, 499494769131, 3761759478289, 28330295424547, 213359105884985, 1606834922893083, 12101280883791521, 91136305878112979, 686359264681956937

```
> reseni(470);
```

$$\left(\frac{8\sqrt{65}}{65}+1\right)\left(\frac{7}{2}+\frac{\sqrt{65}}{2}\right)^{470}+\left(-\frac{8\sqrt{65}}{65}+1\right)\left(\frac{7}{2}-\frac{\sqrt{65}}{2}\right)^{470}$$

```
> simplify(%);
```

26520837159826043939409346283206725510025820002362491501150084834877708 6\
055508527211296286909928777560855021390356541229281115210840279287650216\
574953500030150435861862879337342454857863714815604967429363707208124233\
248039654185923153897500644795380671354542950621310535012883157830604463\
206947728576578090014030228092070884497081629150360510847462768027723748\
4678606566330178357747016612088452470187144990799953 7

```
> convert(%,float);
```

$$0.2652083716\ 10^{413}$$

```
>
```

Příklad:

Kolika způsoby lze pokrýt šachovnici o rozměrech 2 x n



kostkami o rozměru 1 x 2 ?



Řešení:

Označme a_n počet různých pokrytí. Snadno odvodíme, že pro každé přirozené n platí $a_{n+2}=a_{n+1}+a_n$. Dále je jasné, že $a_1=1$ a $a_2=2$. Vyřešme tedy tuto rekurentní rovnici.

```
> rsolve({a(n+2)=a(n+1)+a(n),a(1)=1,a(2)=2},a(n));
```

$$\left(\frac{\sqrt{5}}{10}+\frac{1}{2}\right)\left(\frac{\sqrt{5}}{2}+\frac{1}{2}\right)^{n}+\left(\frac{1}{2}-\frac{\sqrt{5}}{10}\right)\left(-\frac{\sqrt{5}}{2}+\frac{1}{2}\right)^{n}$$

```
> seq(simplify(%),n=1..20);
```

1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233, 377, 610, 987, 1597, 2584, 4181, 6765, 10946

Počty pokrytí tvoří (posunutou) Fibonacciho posloupnost.

```
>
```

Příklad:

Dům má n pater. V každém poschodí bydlí 6 lidí. Vypočtěte, kolik lze vytvořit skupin lidí složených z obyvatel tohoto domu, v nichž se nevyskytují žádní dva lidé bydlící ve stejném poschodí ani ve dvou po sobě jdoucích poschodích.

Řešení:

Označme a_n počet takových skupin. Pak pro každé přirozené n platí rekurentní vztah $a_{n+2} = a_{n+1} + 6\,a_n$. Dále není těžké si promyslet, že $a_1 = 7$ a $a_2 = 13$.

Vyřešme tedy tuto rekurentní rovnici.

```
> reseni:=rsolve({a(n+2)=a(n+1)+6*a(n),a(1)=7,a(2)=13},a(n));
```

$$reseni := -\frac{4\,(-2)^n}{5} + \frac{9\,3^n}{5}$$

Dostali jsme obecný vztah pro hledaný počet skupin lidí.
Pozn: Ve výsledku je zahrnuta i prázdná skupina neobsahující nikoho.

Prvních 20 hodnot:

```
> seq(reseni,n=1..20);
```

7, 13, 55, 133, 463, 1261, 4039, 11605, 35839, 105469, 320503, 953317, 2876335, 8596237, 25854247, 77431669, 232557151, 697147165, 2092490071, 6275373061

```
>
```

Tvorba vlastních balíčků a nápovědy

Ukládání obsahu proměnných:

```
> restart;
> currentdir();
```

"D:\DOKUMENTY\!--PETR--!\VSB-VYUKA\2012-leto (MAMA)\!--PREDNASKY--!\07"

```
> cesta:=cat(currentdir(),"/promenna.m");
```

cesta := "D:\DOKUMENTY\!--PETR--!\VSB-VYUKA\2012-leto (MAMA)\!--PREDNASKY--!\07/promenna.m"

```
> a;
```

$$a$$

```
> a:=2;
```

$$a := 2$$

```
> save(a, cesta);
> restart;
> a;
```

$$a$$

```
> cesta:=cat(currentdir(),"/promenna.m");
```

cesta := "D:\DOKUMENTY\!--PETR--!\VSB-VYUKA\2012-leto (MAMA)\!--PREDNA\
SKY--!\07/promenna.m"

```
> read(cesta);
> a;
```

$$2$$

```
>
```

Tvorba balíčků:
(balíček je tabulka procedur)

```
> restart;
> cesta:=cat(currentdir(),"/balicek.m");
```

cesta := "D:\DOKUMENTY\!--PETR--!\VSB-VYUKA\2012-leto (MAMA)\!--PREDNA\
SKY--!\07/balicek.m"

```
> balicek[soucet]:=proc(a,b)
   a+b;
  end;
```

$$balicek_{soucet} := \mathbf{proc}(a, b)\ a + b\ \mathbf{end\ proc}$$

```
>
> balicek[soucin]:=proc(a,b)
   a*b;
  end;
```

$$balicek_{soucin} := \mathbf{proc}(a, b)\ a*b\ \mathbf{end\ proc}$$

```
>
> balicek[mocnina]:=proc(a,b)
   evalf(a^b);
  end;
```

$$balicek_{mocnina} := \mathbf{proc}(a, b)\ \mathrm{evalf}(\,a\hat{}b\,)\ \mathbf{end\ proc}$$

```
>
> save(balicek, cesta);
> restart;
> libname;
```

"C:\Program Files\Maple 15/lib", "C:\Program Files\Maple 15\toolbox\NAG\lib"

```
> libname:=libname,currentdir();
```

libname := "C:\Program Files\Maple 15/lib",
"C:\Program Files\Maple 15\toolbox\NAG\lib", "D:\DOKUMENTY\!--PETR--!\VSB-VY\
UKA\2012-leto (MAMA)\!--PREDNASKY--!\07"

```
> with(balicek);
```

$$[mocnina, soucet, soucin]$$

```
> mocnina(3,3/7);
```

$$1.601328886$$

```
> soucet(3,3/7);
```

$$\frac{24}{7}$$

```
> soucin(3,3/7);
```

$$\frac{9}{7}$$

```
> restart;
> soucin(3,3/7);
```

$$\mathrm{soucin}\left(3, \frac{3}{7}\right)$$

```
>
```

Tvorba nápovědy:

```
> restart;
> libname:=libname,currentdir();
```

libname := "C:\Program Files\Maple 15\lib",
"C:\Program Files\Maple 15\toolbox\NAG\lib", "D:\DOKUMENTY\!--PETR--!\VSB-VY\UKA\2012-leto (MAMA)\!--PREDNASKY--!\07"

Vytvoříme maplovský zápisník s nápovědou. V menu pak zvolíme "Help - Save to Database".
Vyplníme "Topic" a "Database".
V "moderním" rozhraní zvolíme "Tools - Help Database - Save As Help Page..."

```
> ?napoveda
> restart;
> ?napoveda
>
```

Cvičení

1) Posloupnost (a_n) splňuje (pro každé přirozené n) rekurentní vztah $a_{n+2} = 4\,a_{n+1} - 4\,a_n + n^2$. Dále víme, že $a_1 = 1$ a $a_2 = 4$. Určete obecný předpis pro n-tý člen takovéto posloupnosti. Pokud bude předpis komplikovaný, pokuste se jej zjednodušit.
- Vypište prvních 20 členů naší posloupnosti.
- Najděte obecný vzorec pro součet prvních n členů posloupnosti (a_n). Výsledný vztah opět

upravte.

2) Dům má n pater. V každém poschodí bydlí 4 lidi. Vypočtěte, kolik lze vytvořit skupin lidí složených z obyvatel tohoto domu, v nichž se nevyskytují žádní dva lidé bydlící ve stejném poschodí ani ve dvou po sobě jdoucích poschodích ani ve dvou poschodích, které jsou "ob jedno".
To znamená, že pokud někoho vybereme např. z **5.** patra, nemůže už být nikdo další z pater č. 3, 4, **5**, 6 a 7 vybrán.
- Snažte se výsledek upravit tak, aby neobsahoval žádná imaginární čísla.
- Jaký je nejmenší počet pater potřebných k tomu, aby počet vyhovujících skupin byl větší než 1 000 000 ?

Nápověda: podívejte se, jak jsme řešili podobnou úlohu na přednášce.

3) Nechť a_1 a a_2 jsou libovolně zvolená reálná čísla. Dále nechť pro každé přirozené číslo n platí $a_{n+2} = \frac{a_n + a_{n+1}}{2}$.
Vypočtěte $\lim\limits_{n \to \infty} a_n$.

4) Uvažujme "žebřík" (stejně jako na přednášce) skládající se z n čtverců (viz obrázek níže). Některé z vrcholů čtverců obarvíme. Uděláme to ale tak, aby v každém čtverci byl obarvený alespoň jeden vrchol.
Určete (obecně) počet všech takových obarvení.



5) Jaká je limita $\lim\limits_{n \to \infty} \frac{a_{n+1}}{a_n}$ podílu dvou sousedních členů Fibonacciho posloupnosti 1, 1, 2, 3, 5, 8, 13, 21, 34, 55, ... ?

6) Najděte (obecný) předpis pro n-tou mocninu matice $A = \begin{bmatrix} -4 & -13 \\ 2 & 6 \end{bmatrix}$. Svůj výsledek použijte k výpočtu matice A^{16}. Poté proveďte kontrolu tak, že přímo vypočtete A^{16}.

>